
In this lecture we associate to each permutation puzzle a group, called the puzzle group. We then see that this
group can be represented by a group of permutations, and we can use SAGE to investigate the puzzles.

12.1 Puzzle Groups

Let’s first recall the definition of a permutation puzzle, since we would like to see how groups come into the
picture. In Lecture 1 we defined what we mean by a one person game, and from that we gave the following
definition of a permutation puzzle.

A permutation puzzle is a one person game (solitaire) with a finite set T = {1, 2, . . . , n} of puzzle pieces
satisfying the following four properties:

1. For some n > 1 depending only on the puzzle’s construction, each move of the puzzle corresponds to a
unique permutation of the numbers in T,

2. If the permutation of T in (1) corresponds to more than one puzzle move then the two positions reached
by those two respective moves must be indistinguishable,

3. Each move, say M , must be ”invertible” in the sense that there must exist another move, say M−1, which
restores the puzzle to the position it was at before M was performed, In this sense, we must be able to
“undo” moves.

4. IfM1 is a move corresponding to a permutation τ1 of T and ifM2 is a move corresponding to a permutation
τ2 of T then M1 ·M2 (the move M1 followed by the move M2) is either

• not a legal move, or
• corresponds to the permutation τ1τ2.

As indicated in part 4 it may happen that the composition of two moves is not legal. For example, this happens
with the 15-Puzzle since legal moves change as the empty space moves around the board. See Section 12.5.
This generally happens when dealing with a puzzle that contains a “gap”. We won’t consider such puzzles in
this lecture, besides a remark in Section 12.5. Instead we will focus on puzzles for which two moves can always
be composed. Typically these are the puzzles ”without-gaps”.

Let Puz be a permutation puzzle (where any two moves can be composed). For example Puz could be Rubik’s
cube, Oval Track, or Hungarian Rings. We consider two puzzle moves, m1 and m2, to be equivalent if the two
positions reached by those two respective moves are indistinguishable.

Let M(Puz) be the set of all inequivalent puzzle moves (what we typically refer to a move-sequences). We
can think of M(Puz) as just the set of all possible configurations, or positions of the puzzle pieces. We have
a way to combine elements of M(Puz): if m1,m2 ∈ M then m1m2 represents the move-sequence m1 followed
by m2, which is again in M(Puz). (This is why we assume the puzzle does not have gaps.) It turns out that
M(Puz) is a group under this operation. The identity is the “do nothing” move, and inverses exist by part
3 of the definition above. Associativity follows from the fact that “moves” correspond to “permutations” and
permutation composition is associative.

Definition 12.1 (Puzzle Group) For a permutation puzzle Puz, the set of all inequivalent puzzle movesM(Puz)
is a group under move composition. M(Puz) is called the puzzle group of Puz.

Since puzzle moves and positions correspond to permutations we can represent M(Puz) as a subgroup of a
permutation group. To do this we just need to associate each basic legal move mi ∈ M(Puz), 1 ≤ i ≤ k, to
a permutation αi. We then use the permutation group 〈α1, α2, . . . , αk〉 to represent the puzzle. We’ve already
done this with all of our puzzles, so here we are just emphasizing the fit within group theory.

12.2 Rubik’s Cube

Let Puz be an n× n× n Rubik’s cube, then we call M(Puz) the n-cube group. In the special case when n = 3
we call it the Rubik’s cube group. We use the special notation RCn to denote the n-cube group.

12.2.1 3-Cube Group

We will do a little investigation into the Rubik’s cube group.

As we described in Lecture 1, we label the facets of the Rubik’s Cube as shown Figure 1. Figure 2 shows the
labeling on an actual cube.

The permutation corresponding to each of the basic moves of the Rubik’s Cube are:

R = (25, 27, 32, 30)(26, 29, 31, 28)(3, 38, 43, 19)(5, 36, 45, 21)(8, 33, 48, 24)
L = (9, 11, 16, 14)(10, 13, 15, 12)(1, 17, 41, 40)(4, 20, 44, 37)(6, 22, 46, 35)
U = (1, 3, 8, 6)(2, 5, 7, 4)(9, 33, 25, 17)(10, 34, 26, 18)(11, 35, 27, 19)
D = (41, 43, 48, 46)(42, 45, 47, 44)(14, 22, 30, 38)(15, 23, 31, 39)(16, 24, 32, 40)
F = (17, 19, 24, 22)(18, 21, 23, 20)(6, 25, 43, 16)(7, 28, 42, 13)(8, 30, 41, 11)
B = (33, 35, 40, 38)(34, 37, 39, 36)(3, 9, 46, 32)(2, 12, 47, 29)(1, 14, 48, 27)

R−1, L−1, U−1, D−1, F−1, B−1 correspond to the inverses of these permutations.

Since the centre’s of the cube are fixed by these moves then any two of these moves are inequivalent. This
means that RC3 can be represented by the group subgroup of S48 generated by these permutations:

Figure 1: Facet labeling on the Rubik’s cube.

(a) Labeling on Up, Right, Front faces (b) Labeling on Down, Back, Left faces

Figure 2: The labeling of the facets of Rubik’s Cube.

RC3 = 〈R,L,U,D,F,B〉.

We can define RC3 in SAGE as follows.

SAGE
sage: S48=SymmetricGroup(48)
sage: R=S48("(25,27,32,30)(26,29,31,28)(3,38,43,19)(5,36,45,21)(8,33,48,24)")
sage: L=S48("(9,11,16,14)(10,13,15,12)(1,17,41,40)(4,20,44,37)(6,22,46,35)")
sage: U=S48("(1,3,8,6)(2,5,7,4)(9,33,25,17)(10,34,26,18)(11,35,27,19)")
sage: D=S48("(41,43,48,46)(42,45,47,44)(14,22,30,38)(15,23,31,39)(16,24,32,40)")
sage: F=S48("(17,19,24,22)(18,21,23,20)(6,25,43,16)(7,28,42,13)(8,30,41,11)")
sage: B=S48("(33,35,40,38)(34,37,39,36)(3,9,46,32)(2,12,47,29)(1,14,48,27)")
sage: RC3=S48.subgroup([R,L,U,D,F,B]) # define Rubik’s cube group to be RC3

Now that RC3 is in SAGE we can calculate some facts about the Rubik’s cube. For example, we can determine

the size of RC3. This is the number of different configurations there are of the cube.
SAGE

sage: RC3.order()
43252003274489856000
sage: factor(RC3.order())
2ˆ27 * 3ˆ14 * 5ˆ3 * 7ˆ2 * 11

Therefore there are approximately 4.3 · 1019 configurations of the cube. And only one solution!

Theorem 12.1 The Rubik’s cube group RC3 has order 227314537211 = 43, 252, 003, 274, 489, 856, 000.

Since the order of an element in a group must divide the size of the group, then we immediately see from the
factored form of |RC3| that there are no elements of prime order ≥ 13. Also, by Cauchy’s theorem (see Lecture
11), there must be an element of order 11. Actually finding such an element is another story, all we know is
one exists. In fact, 9 others must exist as well since it would generate a subgroup of order 11.

We can also check if it is possible to flip a single edge, while leaving everything else in place. Consider flipping
the cubie in the uf cubical, the corresponding permutation is (7, 18). The following calculation shows it is not
in RC3.

SAGE
sage: S48("(7,18)") in RC3
False

However, we can flip two edges, say for example the cubies in the uf and ur cubicals. This corresponds to the
permutation (7, 18)(5, 26).

SAGE
sage: S48("(7,18)(5,26)") in RC3
True

Notice this only tells us that it is possible to flip two edges using moves R,L,U,D, F,B, but it doesn’t indicate
what sequence of moves will do this. This is in fact a much harder problem. Basically what we are asking
for is a method which can determine, for any element of RC3, a way to write it as a product of the generators
(or equivalently, as a word in R,L,U,D, F,B). This is known as the word problem in group theory and is very
difficult in many situations.

However, there is an implementation in SAGE of an algorithm for solving the word problem in RC3. It doesn’t
return the shortest possible move sequence, but it does a pretty good job nonetheless. For this we need to use
the built-in CubeGroup() package.

SAGE
sage: rubik=CubeGroup();
sage: G=rubik.group();
sage: R=rubik.R();
sage: L=rubik.L();
sage: U=rubik.U();
sage: D=rubik.D();
sage: F=rubik.F();
sage: B=rubik.B();
sage: state = G("(7,18)(5,26)")
sage: rubik.solve(state) # calls the solve algorithm
"F2 R2 B’ F’ D’ F D B R2 F’ R’ F’ R"

Therefore, one move-sequence for flipping edges uf and ur is

F 2R2B−1F−1D−1FDBR2F−1R−1F−1R.

12.2.2 2-Cube Group

We label the facets of the Pocket Cube as shown in Figure 3. Figure 9 shows the labeling on an actual cube.

Figure 3: Facet labeling on the Pocket cube.

(a) Labeling on Up (blue),
Right (yellow), Front (red)
faces

(b) Labeling on Down
(green), Back (orange),
Left (white) faces

Figure 4: The labeling of the facets of the Pocket Cube.

The permutation corresponding to each of the basic moves of the Pocket Cube are:

R = (13, 14, 16, 15)(10, 2, 19, 22)(12, 4, 17, 24)
L = (5, 6, 8, 7)(3, 11, 23, 18)(1, 9, 21, 20)
U = (1, 2, 4, 3)(9, 5, 17, 13)(10, 6, 18, 14)
D = (21, 22, 24, 23)(11, 15, 19, 7)(12, 16, 20, 8)
F = (9, 10, 12, 11)(3, 13, 22, 8)(4, 15, 21, 6)
B = (17, 18, 20, 19)(1, 7, 24, 14)(2, 5, 23, 16)

R−1, L−1, U−1, D−1, F−1, B−1 correspond to the inverses of these permutations.

There is one major difference between the Pocket cube and Rubik’s cube: the Pocket cube does not have any
fixed centres. Why does this matter? Consider the moves R and L. They are equivalent! Notice that applying
R, leaves the cube in exactly the same position as L (the cube as a whole has just been rotated in space).
Another way to say this is RL−1 is the identity in RC2. Try it!

But if we were to use the permutations above to generate a group then this wouldn’t be the group RC2. Since
the product of permutations associated with R and L don’t have the property that RL−1 = ε. This means the
permutations are picking up the fact that the cube rotated in space.

Again let’s summarize the real difference between Rubik’s cube and the Pocket cube: the Pocket cube can be
rotated in space using only puzzle moves (which rotate faces), whereas Rubik’s cube cannot be rotated in space
using puzzle moves (since centres stay fixed under face rotations).

This means that RC2 is smaller that the permutation group generated by the 6 permutations above. In fact,
we really only need one of each of the following pairs of moves: {R,L}, {U,D}, {F,B}. We’ll choose to only use
R,D,F . This means the UBL cubie always remains in its home position. This is the piece we will keep fixed.

SAGE
sage: S24=SymmetricGroup(24)
sage: R=S24("(13,14,16,15)(10,2,19,22)(12,4,17,24)")
sage: D=S24("(21,22,24,23)(11,15,19,7)(12,16,20,8)")
sage: F=S24("(9,10,12,11)(3,13,22,8)(4,15,21,6)")
sage: RC2=S24.subgroup([R,D,F]) # define Pocket cube group to be RC2

We can determine the size of RC2.

SAGE
sage: RC2.order()
3674160
sage: factor(RC2.order())
2ˆ4 * 3ˆ8 * 5 * 7

Therefore there are approximately 3.6 million configurations of the Pocket cube. And only one solution.

Theorem 12.2 The Pocket cube group RC2 has order 24385 · 7 = 3, 674, 160.

If we didn’t realize that some moves are equivalent, and just constructed the group generated by all moves,
what would happen?

SAGE
sage: S24=SymmetricGroup(24)
sage: R=S24("(13,14,16,15)(10,2,19,22)(12,4,17,24)")
sage: L=S24("(5,6,8,7)(3,11,23,18)(1,9,21,20)")
sage: U=S24("(1,2,4,3)(9,5,17,13)(10,6,18,14)")
sage: D=S24("(21,22,24,23)(11,15,19,7)(12,16,20,8)")
sage: F=S24("(9,10,12,11)(3,13,22,8)(4,15,21,6)")
sage: B=S24("(17,18,20,19)(1,7,24,14)(2,5,23,16)")
sage: S24.subgroup([R,L,U,D,F,B]).order()
88179840
sage: 88179840/3674160
24

We would have been off by a factor of 24. Why 24? This is precisely the number of different rotations there are
for the whole cube. Since the permutation group was treating rotations of the cube as different states, but the
cube group RC2 should know these states really aren’t different at all, then it is no surprise that we would be
off by the number of rotations to the cube: 24.

This does illustrate, however, that we can’t just assign a permutation to each move, and form the permutation
group. Some thought needs to be taken as to whether the representation is faithful.

Swapping Corners on the Pocket Cube:

Are we able to swap two corners on the Pocket Cube, while keeping every other cubie in its home location (not
necessarily with proper orientation)?

If we think about what a typical permutation would look like, well this would be quite tedious. Since corners
can possibly twist and be returned to their home locations, it is not a simply matter of just asking if a 2-cycle
is in RC2. However, we aren’t really interested in how the stickers move around, just the cubies themselves.
So if we view RC2 acting on the the 8 cubies, we just want to know if we can swap two cubies, and fix all other
cubies in their current location.

If we number the cubicles as follows: 1 is the ufr cubical, 2 is the urb cubical, 3 is the ubl cubical, 4 is the ulf
cubical, 5 is the dfr cubical, 6 is the drb cubical, 7 is the dbl cubical, 8 is the dlf cubical.

The action of each move on the cubies are then:

R = (1, 2, 6, 5)
L = (3, 4, 8, 7)
U = (1, 4, 3, 2)
D = (5, 6, 7, 8)
F = (1, 5, 8, 4)
B = (2, 3, 7, 6)

We can the ask SAGE to compute whether it is possible to swap the 1 and 2 cubies.

SAGE
sage: S8=SymmetricGroup(8)
sage: R=S8("(1,2,6,5)")
sage: L=S8("(3,4,8,7)")
sage: U=S8("(1,4,3,2)")
sage: D=S8("(5,6,7,8)")
sage: F=S8("(1,5,8,4)")
sage: B=S8("(2,3,7,6)")
sage: H=S8.subgroup([R,L,U,D,F,B])
sage: S8("(1,2)") in H
True
sage: H.order()=factorial(8)
True

The computation shows that not only can we swap cubies 1 and 2, but in fact every permutation of the 8 cubies
is possible. Remember though, the representation of RC2 that we chose to work with here ignores any twisting
of corners. So even though we can move the pieces anywhere we want, there may be limitations on how we can
twist them.

12.3 Oval Track

Let Puz be the Oval Track puzzle (or one of its variations), then we call M(Puz) the Oval Track group and
we use the special notation OT to denote this group.

We’ll look at a few different variations of the puzzle, corresponding to different modifications of the turntable
move T .

12.3.1 Oval Track - TopSpin: T = (1, 4)(2, 3)

The basic legal moves of the TopSpin version of the Oval Track puzzle are R, and T , where R denotes a
clockwise rotation of numbers around the track, where each number moves one space, and T denotes a rotation
of the turntable. See Figure 5.

Figure 5: The Oval Track Puzzle.

The permutation corresponding to the legal moves R, and T are as follows:

R = (1, 2, 3, 4, 5, 6, 7, 8, 9, 10, 11, 12, 13, 14, 15, 16, 17, 18, 19, 20)
T = (1, 4)(2, 3)

Note that T−1 = T . This is due to the fact that spinning the turntable in either direction achieves the same
result.

The basic moves R and T are not equivalent, so OT can be represented by the permutation group generated by
these two permutations.

SAGE
sage: S20=SymmetricGroup(20)
sage: R=S20("(1,2,3,4,5,6,7,8,9,10,11,12,13,14,15,16,17,18,19,20)")
sage: T=S20("(1,4)(2,3)")
sage: OT=S20.subgroup([R,T]) # define OT to be a permutation group

What is the size of OT . Since the puzzle consists of permuting 20 disks, we wonder if all permutations are
possible. Since there are 20! permutations of 20 objects, we’d like to know if |OT | = 20!.

SAGE
sage: OT.order()==factorial(20)
True

This means OT is actually the symmetric group of degree 20: OT = S20. Therefore, every permutation of the
disks is possible. Of course, the key to solving this puzzle is to figure out how you can obtain each permutation
using only moves R and T .

12.3.2 Oval Track - Variation 2: T = (1, 4, 3, 2)

The turntable move in the original TopSpin puzzle is now replaced with the move indicated by the purple
dashed lines. In this version, the new turntable move for the puzzle in Figure 6 moves the disk in spot 4 to spot
3, the disk in spot 3 to spot 2, the disk in spot 2 to spot 1, and takes the disk in spot 1 to spot 4.

The permutation corresponding to the legal moves R, and T are as follows:

R = (1, 2, 3, 4, 5, 6, 7, 8, 9, 10, 11, 12, 13, 14, 15, 16, 17, 18, 19, 20)
T = (1, 4, 3, 2)

Figure 6: The Oval Track Puzzle.

SAGE
sage: S20=SymmetricGroup(20)
sage: R=S20("(1,2,3,4,5,6,7,8,9,10,11,12,13,14,15,16,17,18,19,20)")
sage: T=S20("(1,4,3,2)")
sage: OT2=S20.subgroup([R,T]) # define OT2 to be a permutation group
sage: OT2.order()==factorial(20)
True

In this variation all possible permutations of the 20 disks are possible.

12.3.3 Oval Track - Variation 3: T = (1, 6)(2, 5)(3, 4)

Another version of the turntable move involving 6 disks is given in Figure 7.

Figure 7: The Oval Track Puzzle.

SAGE
sage: S20=SymmetricGroup(20)
sage: R=S20("(1,2,3,4,5,6,7,8,9,10,11,12,13,14,15,16,17,18,19,20)")
sage: T=S20("(1,6)(2,5)(3,4)")
sage: OT3=S20.subgroup([R,T]) # define OT3 to be a permutation group
sage: OT3.order()==factorial(20)
True

In this variation all possible permutations of the 20 disks are possible.

12.4 Hungarian Rings

Let Puz be the Hungarian Rings puzzle (numbered version), then we call M(Puz) the Hungarian Rings
group and we use the special notation HR to denote this group.

The basic legal moves of the Hungarian Rings puzzle are R, and L, where R denotes a clockwise rotation of
numbers around the right-hand ring (each number moves one space), and L denotes a clockwise rotation of
numbers around the left-hand ring.

Figure 8: Hungarian Rings - numbered version.

The permutation corresponding to each of the legal moves R and L are:

R = (1, 38, 37, 36, 35, 6, 34, 33, 32, 31, 30, 29, 28, 27, 26, 25, 24, 23, 22, 21)
L = (1, 20, 19, 18, 17, 16, 15, 14, 13, 12, 11, 10, 9, 8, 7, 6, 5, 4, 3, 2)

R−1 and L−1 correspond to the inverses of these permutations.

Since the moves R and L are inequivalent then HR can be represented by the group of permutations generated
by R and L.

SAGE
sage: S38=SymmetricGroup(38)
sage: L=S38("(1,20,19,18,17,16,15,14,13,12,11,10,9,8,7,6,5,4,3,2)")
sage: R=S38("(1,38,37,36,35,6,34,33,32,31,30,29,28,27,26,25,24,23,22,21)")
sage: HR=S38.subgroup([R,L])==factorial(38)
True

Therefore, all possible permutations of the 38 balls are possible.

12.5 15-Puzzle

The 15-puzzle does not fit into group theory as neatly as our other puzzles do. The problem is that a move must
involve the empty space, so the available legal moves at each stage changes depending on where the empty
space is.

In what follows, we will describe a move of the pieces of the 15 puzzle by the first letter of the word (u)p, (d)own,
(l)eft, (r)ght, which is to indicate the direction a tile is pushed into the empty space. For example, beginning
with the empty space in spot 16, let m1 be the sequence of moves:

m1 = rrr.

Similarly, with the empty space in spot 16, let m2 be the sequence of moves:

m2 = rddd.

Move m1 places the empty space in spot 13 by moving all tile on the bottom row to the right. Whereas, move m2

places the empty space in spot 3. Therefore, it is impossible to perform the move sequence m1m2 = (rrr)(rddd)
since once three r moves are applied there is no tile to the left of the empty space to apply another r move. The
set of all legal moves is not closed under composition, therefore is not a group.

However, if we narrow our focus we can find a group lurking in there somewhere.

Represent each sequence of moves by its corresponding permutation, so the set of all such move sequence
corresponds to a subset of the permutation group S16. Let this subset be denoted by FP :

FP = {α | α is the permutation corresponding to a legal position of the 15-puzzle}.

We already noted FP is not a group but the example gives us some insight into how we can fix this. If each
moves starts with the empty space in box 16, then returns it to box 16, then the next move can be applied
without any trouble. We let FP ∗ consist of the set of all moves that leaves the empty space in spot 16. In terms
of permutations this means:

FP ∗ = {α ∈ FP | α(16) = 16}.

Now FP ∗ is a group. In fact we know it to be the group A15.

In general when considering puzzles with gaps, we can look at the subset of legal moves where each move
returns the space to its home position, this set will form a group.

12.6 Exercises

1. Single Corner Twist. Is it possible to rotate a single corner cubie of Rubik’s cube, while leaving every-
thing else in its home position?

(a) Figure for Exer-
cise 1

(b) Figure for Exer-
cise 2

(c) Figure for Exer-
cise 3

Figure 9: Which corner twists are possible?

2. Two Corner Twists. For Rubik’s cube, is it possible to rotate two corner cubies in the same direction,
while leaving everything else in its home position?

3. Another Two Corner Twists. For Rubik’s cube, is it possible to rotate two corner cubies in opposite
directions, while leaving everything else in its home position?

4. Swapping Corners on Rubik’s cube. Show that it is impossible to swap two corner cubies on Rubik’s
cube, while leaving all other cubies in their home locations (not necessarily with proper orientation)?

5. Oval Track with 19 Disks. Consider the Oval Track puzzle (TopSpin version) where only 19 disks are
used. Are all permutations of the 19 disks possible? If not, can you describe exactly which permutation
are possible?

6. Varying the Number of Disks on Oval Track. For the Oval Track puzzle with n disk, let OTn denote
the puzzle group, determine the size of OTn, for 6 ≤ n ≤ 20. In each case, describe exactly which
permutations of the puzzle pieces are possible.

7. Very Few Disks on Oval Track. Consider OTn for n = 4, 5. Investigate which permutations of the
puzzle pieces are possible.

8. Varying the turntable move T of the Oval Track puzzle. In this exercise you will investigate, with
the help of SAGE , some variations of the Oval Track puzzle. In all variations1, we assume there are 20
disks, and the usual move consisting of rotating the pieces along the track isR. We will vary the turntable
move T . We have already seen that if the turntable move is T = (1, 4)(2, 3) or T = (4, 3, 2, 1) then we are
still able to obtain all permutations of the 20 disks. Investigate the other variations of the move T given
in Table 1. Under the column “permutation group”, try to determine what groups of permutation of the
20 pieces is possible. The first two rows have been filled in already.

variation turntable move T permutation group

OT 1 (1, 4)(2, 3) S20

OT 2 (4, 3, 2, 1) S20

OT 3 (3, 2, 1)
OT 4 (5, 4, 3, 2, 1)
OT 5 (1, 2)(3, 4)
OT 6 (1, 11)(4, 14)
OT 7 (5, 3, 1)
OT 8 (1, 3)(2, 4)

Table 1: Variations of the Oval Track puzzle

1Variation names are due to John O. Kiltinen who studies these in his book: Oval Track and other Permutation Puzzles.

